Higher-order functions
Informatics 1 – Functional Programming: Tutorial 3

Heijltjes, Wadler

Due: The tutorial of week 5 (22/23 Oct.)

Please attempt the entire worksheet in advance of the tutorial, and bring with you all work, including (if a computer is involved) printouts of code and test results. Tutorials cannot function properly unless you do the work in advance.

You may work with others, but you must understand the work; you can't phone a friend during the exam.

Assessment is formative, meaning that marks from coursework do not contribute to the final mark. But coursework is not optional. If you do not do the coursework you are unlikely to pass the exams.

Attendance at tutorials is obligatory; please let your tutor know if you cannot attend.

Higher-order functions

Haskell functions are *values*, which may be processed in the same way as other data such as numbers, tuples or lists. In this tutorial we'll use a number of *higher-order functions*, which take other functions as arguments, to write succinct definitions for the sort of list-processing tasks that you've previously coded explicitly using recursion or comprehensions.

The first part of the tutorial deals with three higher-order functions, `map`, `filter`, and `fold`. For each of these you will be asked to write three functions. The second part deals with `fold` in some more detail, and will ask you to write functions using both `map` and `filter` at the same time.

Map

Transforming every list element by a particular function is a common need when processing lists—for example, we may want to

- add one to each element of a list of numbers,
- extract the first element of every pair in a list,
- convert every character in a string to uppercase, or
- add a grey background to every picture in a list of pictures.

The `map` function captures this pattern, allowing us to avoid the repetitious code that results from writing a recursive function for each case.

Consider a function `g` defined in terms of an imaginary function `f` as follows:

```
g []     = []
g (x:xs) = f x : g xs
```

The function `g` can be written with recursion (as above), or with a comprehension, or with `map`: all three definitions are equivalent.



Figure 1: The `map` function

```
g xs  =  [ f x | x <- xs ]
g xs  =  map f xs
```

Below right is the definition of `map`. Note the similarity to the recursive definition of `g` (below left). As compared with `g`, `map` takes one additional argument: the function `f` that we want to apply to each element.

```
g []     = []
g (x:xs) = f x : g xs
```

```
map :: (a -> b) -> [a] -> [b]
map f []     = []
map f (x:xs) = f x : map f xs
```

Given `map` and a function that operates on a single element, we can easily write a function that operates on a list. For instance, the function that extracts the first element of every pair can be defined as follows (using `fst :: (a,b) -> a`):

```
fsts :: [(a,b)] -> [a]
fsts pairs = map fst pairs
```

Exercises

1. Using `map` and other suitable library functions, write definitions for the following:
   (a) A function `uppers :: String -> String` that converts a string to uppercase.
   (b) A function `doubles :: [Int] -> [Int]` that doubles every item in a list.
   (c) A function `penceToPounds :: [Int] -> [Float]` that turns prices given in pence into the same price in pounds.
   (d) Write a list-comprehension version of `uppers` and use it to check your answer to (a).

Filter

Removing elements from a list is another common need. For example, we might want to remove non-alphabetic characters from a string, or negative integers from a list. This pattern is captured by the `filter` function.

Consider a function `g` defined in terms of an imaginary predicate `p` as follows (a predicate is just a function into a `Bool` value):

```
g []     = []
g (x:xs) | p x       = x : g xs
         | otherwise = g xs
```

The function `g` can be written with recursion (as above), or with a comprehension, or with `filter`: all three definitions are equivalent.

```
g xs  =  [ x | x <- xs, p x ]
g xs  =  filter p xs
```

For instance, we can write a function `evens :: [Int] -> [Int]`, which removes all odd numbers from a list using `filter` and the standard function `even :: Int -> Int`:

```
evens list = filter even list
```

This is equivalent to:

```
evens list = [x | x <- list, even x]
```

Below right is the definition of `filter`. Note the similarity to the way `g` is defined (below left). As compared with `g`, `filter` takes one additional argument: the predicate that we use to test each element.

```
g []     = []
g (x:xs) | p x       = x : g xs
         | otherwise = g xs
```

```
filter :: (a -> Bool) -> [a] -> [a]
filter p []     = []
filter p (x:xs) | p x       = x : filter p xs
                | otherwise = filter p xs
```

Exercises

2. Using `filter` and other standard library functions, write definitions for the following:

   (a) A function `alphas :: String -> String` that removes all non-alphabetic characters from a string.

   (b) Define a function `rmChar :: Char -> String -> String` that removes all occurrences of a character from a string.

   (c) A function `above :: Int -> [Int] -> [Int]` that removes all numbers less than or equal to a given number.

   (d) A function `unequals :: [(Int,Int)] -> [(Int,Int)]` that removes all pairs `(x,y)` where `x == y`.

   (e) Write a list-comprehension version of `rmChar` and use QuickCheck to test it against the version using filter.

Comprehensions, map and filter

As we have seen, list comprehensions process a list using transformations similar to `map` and `filter`. In general, `[f x | x <- xs, p x]` is equivalent to `map f (filter p xs)`.

Exercises

3. Write expressions equivalent to the following using `map` and `filter`. Use QuickCheck to verify your answers.

   (a) `[toUpper c | c <- s, isAlpha c]`

   (b) `[2 * x | x <- xs, x > 3]`

   (c) `[reverse s | s <- strs, even (length s)]`

Fold

The `map` and `filter` functions act on elements individually; they never combine one element with another.

Sometimes we want to combine elements using some operation. For example, the `sum` function can be written like this:

```
sum []     = 0
sum (x:xs) = x + (sum xs)
```

Here we're essentially just combining the elements of the list using the `+` operation. Another example is `reverse`, which reverses a list:

```
reverse []     = []
reverse (x:xs) = reverse xs ++ [x]
```

This function is just combining the elements of the list, one by one, by appending them onto the end of the reversed list. This time the "combining" function is a little harder to see. It might be easier if we wrote it this way:

```
reverse []     = []
reverse (x:xs) = x `snoc` (reverse xs)

snoc x xs = xs ++ [x]
```

Now you can see that `` `snoc` `` plays the same role as `+` played in the example of `sum`.

These examples (and many more) follow a pattern: we break down a list into its head (`x`) and tail (`xs`), recurse on `xs`, and then apply some function to `x` and the modified `xs`. The only things we need to specify are the function (such as `(+)` or `snoc`) and the *initial value* (such as `0` in the case of `sum` and `[]` in the case of `reverse`.

This pattern is called "a fold" and is implemented in Haskell via the function `foldr`.

```
g []     = u
g (x:xs) = x `f` g xs
```

```
foldr :: (a -> b -> b) -> b -> [a] -> b
foldr f u []     = u
foldr f u (x:xs) = x `f` foldr f u xs
```

The function `g` can be written with recursion (as above) or by using a fold: both definitions are equivalent.

```
g xs = foldr f u xs
```

One way to visualize the action of `foldr` is shown in Figure 2. Given a function `f :: a -> b -> b`, an initial value `u :: b` (sometimes called the "unit"), and a list `list :: [a]`, the `foldr` function returns the value that results from replacing every `:` (cons) in `list` with `f` and replacing the terminating `[]` (nil) with `u`.



Figure 2: The `foldr` function

For example, we can define `sum :: [Int] -> Int` as follows, using `(+)` as the function and `0` as the initial value (unit):

```
sum :: [Int] -> Int
sum ns = foldr (+) 0 ns
```

(**Note:** to treat an infix operator like `+` as a function name, we need to wrap it in parentheses.)



Figure 3: Illustration of `foldr (+) 0 [10,20,30]`

Exercises

4. We will practice the use of `foldr` by writing several functions first with recursion, and then using `foldr`. You can use other standard library functions as well. For each pair of functions that you write, test them against each other using QuickCheck.

   (a) Look at the recursive function `productRec :: [Int] -> Int` that computes the product of the numbers in a list, and write an equivalent function `productFold` using `foldr`.

   (b) Write a recursive function `andRec :: [Bool] -> Bool` that checks whether every item in a list is `True`. Then, write the same function using `foldr`, this time called `andFold`.

   (c) Write a recursive function `concatRec :: [String] -> String` that puts a list of strings together to form a single string. Then, write a similar function `concatFold` using `foldr`.

   Note: these functions are similar to the library functions `product`, `and` and `concat`, although the prelude `concat` has the more general type `[[a]] -> [a]`.

   (d) Write a recursive function `rmCharsRec :: String -> String -> String` that removes all characters in the first string from the second string, using your function `rmChar` from exercise (2b).

   ```
   *Main> rmCharsRec ['a'..'l'] "football"
   "oot"
   ```

   Then, write a second version `rmCharsFold` using `rmChar` and `foldr`. Check your functions with QuickCheck.

Optional Material

Matrix manipulation

Next, we will look at matrix addition and multiplication. As matrices we will use lists of lists of `Int`s; for example:

$$\begin{pmatrix} 1 & 4 & 9 \\ 2 & 5 & 7 \end{pmatrix} \quad \text{is represented as} \quad \texttt{[[1,4,9],}\\ \texttt{ [2,5,7]]}$$

The declaration below, which you can find in your `tutorial3.hs`, makes the type `Matrix` a shorthand for the type `[[Int]]`.

```
type Matrix = [[Int]]
```

Our first task is to write a test to show whether a list of lists of `Int`s is a matrix. This test should verify two things: 1) that the lists of `Int`s are all of equal length, and 2) that there is at least one row and one column in the list of lists.

Exercises

5. (a) Write a function `uniform :: [Int] -> Bool` that tests whether the integers in a list are all equal. You can use the library function `all`, which tests whether all the elements of a list satisfy a predicate; check the type to see how it is used. If you want, you can try to define `all` in terms of `foldr` and `map`.

   (b) Using your function `uniform` write a function `valid :: Matrix -> Bool` that tests whether a list of lists of `Int`s is a matrix (it should test the properties 1) and 2) specified above).

Two other things we would like to know is the size (or dimensions) of a matrix, and whether it is square. A useful function here is `uncurry`, which turns a function that takes two arguments into a function that operates on a pair.

Exercises

6. (a) Look up the definition of `uncurry`. What is returned by the following expression?

   ```
   Main> uncurry (+) (10,8)
   ```

   (b) Write a function `size :: Matrix -> (Int,Int)` that returns the number of rows and columns of a matrix as a pair (*rows*, *columns*).

   ```
   Main> size [[1,2,3],[4,5,6]]
   (2,3)
   ```

   (c) Write a function `square :: Matrix -> Bool` that tests whether a matrix is square.

Another useful higher-order function is `zipWith`. It is a lot like the function `zip` that you have seen, which takes two lists and combines the elements in a list of pairs. The difference is that instead of combining elements as a pair, you can give `zipWith` a specific function to combine each two elements. The definition is as follows (Figure 4 gives an illustration):

```
zipWith f [] _ = []
zipWith f _ [] = []
zipWith f (x:xs) (y:ys) = f x y : zipWith f xs ys
```

Exercises

7. Show how to define `zipWith` in terms of `map`, `zip` and `uncurry`.



Figure 4: Illustration of `zipWith` for lists of equal length.

Adding two matrices of equal size is done by pairwise adding the elements that are in the same position, i.e. in the same column and row, to form the new element at that position. For example:

$$\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{pmatrix} + \begin{pmatrix} 10 & 20 & 30 \\ 40 & 50 & 60 \end{pmatrix} = \begin{pmatrix} 11 & 22 & 33 \\ 44 & 55 & 66 \end{pmatrix}$$

We will use `zipWith` to implement matrix addition.

Exercises

8. Write a function `plusM` that adds two matrices. Return an error if the input is not suitable. It might be helpful to define a helper function `addRow` that adds two rows of a matrix.

For matrix multiplication we need what is called the *inner product* or *dot product* of two vectors:

$$(a_1, a_2, \ldots, a_n) \cdot (b_1, b_2, \ldots, b_n) = a_1b_1 + a_2b_2 + \ldots + a_nb_n$$

Matrix multiplication is then defined as follows: two matrices with dimensions (n, m) and (m, p) are multiplied to form a matrix of dimension (n, p) in which the element in row i, column j is the inner product of row i in the first matrix and column j in the second. For example:

$$\begin{pmatrix} 1 & 10 \\ 100 & 10 \end{pmatrix} \times \begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix} = \begin{pmatrix} 31 & 42 \\ 130 & 240 \end{pmatrix}$$

(For more information see `http://en.wikipedia.org/wiki/Matrix_multiplication`.)

Exercises

9. Define a function `timesM` to perform matrix multiplication. Return an error if the input is not suitable. It might be helpful to define a helper function `innerProduct` for the inner product of two vectors (lists).

   Hint: Write a helper function `multRow :: [Int] -> Matrix -> [Int]` that multiplies a single row of a matrix with another matrix. Your function should take the inner product of the single row with every column of the matrix, and return the values as a list. To make the columns of a matrix readily available you can use the function `transpose`.

10. For a real challenge, you can try to compute the inverse of a matrix. There are a few steps involved in this process:

    (a) The entries of a matrix should be changed to `Double`s or (even better) `Rational`s to allow proper division.

    (b) You will need a function to find the *determinant* of a matrix. This will tell you if it has an inverse.

    (c) You will need a function to do the actual inversion.

There are several different algorithms available to compute the determinant and the inverse of a matrix. Good places to start looking are:

`http://mathworld.wolfram.com/MatrixInverse.html`

`http://en.wikipedia.org/wiki/Invertible_matrix`

For this last exercise, no sample solution will be made available. If you have made it this far, you will probably know how to test your solution yourself.